Здравствуйте, знаю что вопрос уже обсуждался, но никак не могу исправить ошибку:

```1c
    {ОбщийМодуль.УправлениеПечатью.Модуль(1522)}: В обработчике печати не был сформирован табличный документ для: ТабельАО
    ВызватьИсключение(ТекстСообщенияОбОшибке);
```

Искал в интернете, в том числе на инфостарте решения, но ничего не помогает.
В итоге плюнул, скопировал код с работающей ВПФ (даже пытался переделать работающую) но все равно та же проблема.

сам тестовый код:

```1c
    Функция СведенияОВнешнейОбработке() Экспорт

    ПараметрыРегистрации = Новый Структура;
    ПараметрыРегистрации.Вставить("Вид", "ПечатнаяФорма");
    ПараметрыРегистрации.Вставить("Назначение", ПолучитьНазначениеОбработки());
    ПараметрыРегистрации.Вставить("Наименование", "Табель тест ");
    ПараметрыРегистрации.Вставить("Версия", "1.4");
    ПараметрыРегистрации.Вставить("Информация", "Табель ");
    ПараметрыРегистрации.Вставить("БезопасныйРежим", Ложь);
    ПараметрыРегистрации.Вставить("ВерсияБСП", "1.2.1.4");

    Команды = ПолучитьТаблицуКоманд();
    ДобавитьКоманду(Команды, "Табель тестовый - АО",
    "ТабельАО",
    "ВызовСерверногоМетода",
    Ложь,
    "ПечатьMXL");

    ПараметрыРегистрации.Вставить("Команды", Команды);

    Возврат ПараметрыРегистрации;

    КонецФункции
    // ТУТ пишем объекты, к которым оотносится эта форма...
    Функция ПолучитьНазначениеОбработки()

    Массив = Новый Массив;
    Массив.Добавить("Документ.ВедомостьНаВыплатуЗарплатыВКассу");

    Возврат Массив;

    КонецФункции

    // Не трогаем....
    Функция ПолучитьТаблицуКоманд()

    Команды = Новый ТаблицаЗначений;
    Команды.Колонки.Добавить("Представление", Новый ОписаниеТипов("Строка"));
    Команды.Колонки.Добавить("Идентификатор", Новый ОписаниеТипов("Строка"));
    Команды.Колонки.Добавить("Использование", Новый ОписаниеТипов("Строка"));
    Команды.Колонки.Добавить("ПоказыватьОповещение", Новый ОписаниеТипов("Булево"));
    Команды.Колонки.Добавить("Модификатор", Новый ОписаниеТипов("Строка"));

    Возврат Команды;

    КонецФункции

    // НЕ трогаем..
    Процедура ДобавитьКоманду(ТаблицаКоманд, Представление, Идентификатор, Использование, ПоказыватьОповещение=Ложь, Модификатор=")

    НоваяКоманда = ТаблицаКоманд.Добавить();
    НоваяКоманда.Представление = Представление;
    НоваяКоманда.Идентификатор = Идентификатор;
    НоваяКоманда.Использование = Использование;
    НоваяКоманда.ПоказыватьОповещение = ПоказыватьОповещение;
    НоваяКоманда.Модификатор = Модификатор;

    КонецПроцедуры

    Процедура Печать(МассивОбъектов, КоллекцияПечатныхФорм, ОбъектыПечати, ПараметрыВывода) Экспорт

    УправлениеПечатью.ВывестиТабличныйДокументВКоллекцию(КоллекцияПечатныхФорм, "ТабельАО", "Табель - АО ", СформироватьПечатнуюФорму(МассивОбъектов,ОбъектыПечати));

    КонецПроцедуры // Печать()

    Функция СформироватьПечатнуюФорму(МассивОбъектов,ОбъектыПечати) экспорт

    ТабличныйДокумент = Новый ТабличныйДокумент;
    ТабличныйДокумент.ИмяПараметровПечати = "ТабельАО";

    МакетОбработки = ПолучитьМакет("МакетТабеля");

    ОбластьШапка = МакетОбработки.ПолучитьОбласть("Шапка");

    ТабличныйДокумент.Вывести(ОбластьШапка);

    ОбластьСтрока = МакетОбработки.ПолучитьОбласть("Строка");

    ТабличныйДокумент.Вывести(ОбластьСтрока);

    ОбластьПодвал = МакетОбработки.ПолучитьОбласть("Подвал");

    ТабличныйДокумент.Вывести(ОбластьПодвал);

    КонецФункции
```
---
```1c
    Функция СформироватьПечатнуюФорму(МассивОбъектов,ОбъектыПечати) экспорт

    ТабличныйДокумент = Новый ТабличныйДокумент;
    ТабличныйДокумент.ИмяПараметровПечати = "ТабельАО";

    МакетОбработки = ПолучитьМакет("МакетТабеля");

    ОбластьШапка = МакетОбработки.ПолучитьОбласть("Шапка");

    ТабличныйДокумент.Вывести(ОбластьШапка);

    ОбластьСтрока = МакетОбработки.ПолучитьОбласть("Строка");

    ТабличныйДокумент.Вывести(ОбластьСтрока);

    ОбластьПодвал = МакетОбработки.ПолучитьОбласть("Подвал");

    ТабличныйДокумент.Вывести(ОбластьПодвал);

    Возврат ТабличныйДокумент; // Это добавь
    КонецФункции
```
---
up
---
```1c
    Функция СформироватьПечатнуюФорму(МассивОбъектов,ОбъектыПечати) экспорт

    ТабличныйДокумент = Новый ТабличныйДокумент;
    ТабличныйДокумент.ИмяПараметровПечати = "ТабельАО";

    МакетОбработки = ПолучитьМакет("МакетТабеля");

    ОбластьШапка = МакетОбработки.ПолучитьОбласть("Шапка");

    ТабличныйДокумент.Вывести(ОбластьШапка);

    ОбластьСтрока = МакетОбработки.ПолучитьОбласть("Строка");

    ТабличныйДокумент.Вывести(ОбластьСтрока);

    ОбластьПодвал = МакетОбработки.ПолучитьОбласть("Подвал");

    ТабличныйДокумент.Вывести(ОбластьПодвал);

    Возврат ТабличныйДокумент; // Это добавь
    КонецФункции
```
---
ТабличныйДокумент; Блин, сильно затупил, спасибо)